NASA TM-85754

NASA Technical Memorandum 85754

NASA-TM-85754 19840009492

A NONINTRUSIVE NUCLEAR MONITOR FOR MEASURING LIQUID CONTENTS IN SEALED VESSELS

JAG J. SINGH AND GERALD H. MALL

FEBRUARY 1984

NASA

National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665

SUMMARY

This report describes a nonintrusive nuclear technique for monitoring fluid contents in sealed vessels, regardless of the fluid distribution inside the vessels. The technique is applicable to all-g environments. It is based on the differences in Cesium-137 gamma ray attenuation coefficients in air and the test liquids.

INTRODUCTION

We recently developed a nuclear technique[1] for monitoring water levels in pressurized vessels. It was based on the differences in attenuation coefficients of water and air for Cs^{137} (662 Kev) gamma rays. It is now proposed that the nuclear monitor can be adapted for measuring the total fluid content in closed vessels regardless of the fluid location inside the vessels. Such a monitor would be useful for 0-g environment or other situations where effective accelerations could be different from that on earth. However, such a monitor would require specialized source and detector geometries.

In this paper, we describe a system for monitoring the arbitrarily distributed liquid content of a vessel and theoretically demonstrate its effectiveness using a computer simulation.

CONFIGURATIONAL DETAILS

The principle of operation of the standard nuclear monitor has been described in detail in reference 1. Briefly, it depends on the fact that

$$I_t(\text{Air}) >> I_t(\text{Liquid})$$

where $$I_t(\text{Air}) = I_o\, e^{-(2\mu_c x_c + \mu_a x_a)} \tag{1}$$

and $$I_t(\text{Liquid}) = I_o\, e^{-(2\mu_c x_c + \mu_\ell x_\ell)} \tag{2}$$

where I_o = Incident Gamma Ray Intensity

μ_c = Linear Attenuation Coefficient for the Container Material

μ_a = Linear Attenuation Coefficient for Air

μ_ℓ = Linear Attenuation Coefficient for the Test Liquid

x_c = Container Wall Thickness

N84-17560 #

x_a = Gamma Ray Pathlength in Air

x_ℓ = Gamma Ray Pathlength in Liquid

The system in reference 1 utilizes a collimated source of gamma rays measured by a detector having wide acceptance angle. This approach is suitable for the more conventional applications where the fluid seeks a stabilized position in the vessel.

This paper considers a redesigned system such that the monitor would work regardless of the fluid's location/distribution inside the container. Several source and detector configurations were considered(*). Some of them are listed below:

1. Uniformly distributed, highly collimated point sources and detectors on opposing faces of the container.

2. Uniform linear source on one face with point detectors uniformly distributed on the opposing face of the container.

3. Non-uniform linear source on one face with a uniform planar detector on the opposite face.

4. Uniform linear source, with various degrees of collimation, on one face with a uniform planar detector on the opposite face of the container.

5. Uniform planar source on one face with a uniform planar detector on the opposite face of the container.

After trying various source/detector arrangements, two major modifications to the scheme in reference 1 were made:

1. The well-collimated point radiation source was replaced by a planar source uniformly deposited on one side of the container. (It was observed that the source could just as well be in the form of a narrow linear strip passing vertically through the center of one side of the container, if the fluid were to be distributed randomly inside the container. However, if the fluid were pressed against one of the walls or confined to corner regions due to centrifugal forces, it would be preferrable to have a planar source covering the entire side.)

2. The second modification required replacing the finite-size cylindrical detector by a planar detector covering the entire opposite wall of the container.

Figure 1 shows a schematic diagram of the proposed concept. The computational details for testing the system are described in the following section.

COMPUTATIONAL PROGRAM

The computer program CRANK is written in FORTRAN IV language for the Control Data CYBER 170 series digital computer system with Network Operating

(*) For the sake of specificity, we assume the container to be a $\ell \times 2d \times 2d$ rectangular box.

System (NOS) 1.4. The program requires approximately 52,000 octal locations of core storage . A typical case in which 40,000 trajectories are computed requires approximately 470 CPU seconds on CYBER 175.

The program models the stochastic process in which a gamma ray traverses a medium subject to multiple Compton scattering[2]. The technique used for tracking the progress of each gamma ray through the medium has been reported in the literature[3,4] and described in detail in reference 1. The medium modeled by the program is representative of a closed container holding variable amounts of a fluid located randomly within the container.

The interior of the container is illustrated in figure 2. The region, of exterior dimensions $\ell \times 2d \times 2d$, is subdivided into smaller volumes such that the cross section in any plane parallel to the x-z plane is an $n \times n$ uniform grid. The parameters ℓ, d, and n are defined by the user as program input. Within this grid, the program assigns to, and identifies each cell with, an integer index between 1 and the total number of cells. If a cell is designated to contain fluid, the electron density (and thus the scattering properties) for that cell is assigned a value corresponding to the fluid. All cells are initially assigned an electron density corresponding to air, 3.65×10^{20} electrons/cm^3.

As currently configured, the program is subject to several limitations, each of which can be changed with minor modifications to the code. The shape of the container is assumed to be a parallelopiped with a square cross section in the x-y plane. The initial position of each gamma ray is assumed to be uniformly distributed in the x-y plane and to be contained within the container interior (i.e. $|x| \leqslant d$, $|y| \leqslant d$). Each gamma ray crossing the $z = \ell$ plane within the container interior is assumed to be counted.

Program execution is performed in sets where the number of sets is provided by the user. A single test contains repetitions for several fluid distributions where the number of distributions is a function of the grid selected by the user. For a 10 x 10 grid, 11 distributions are considered corresponding to fluid content of 0, 10, 20, ..., 90, 100%. For a 5 x 5 grid, 6 distributions are considered corresponding to fluid content of 0, 20, 40, 60, 80, 100%. For each distribution the number of gamma ray paths to be traced is specified by the user.

For each gamma ray path to be traced the program first assigns an initial gamma ray position and direction. The initial Z coordinate of the particle (ZI) is set to zero. The X and Y coordinates (XI, YI) are assigned uniform random values between -d and d. The initial direction of the particle is defined by the standard spherical coordinates θ and ϕ where the initial value of θ is assigned a uniform random value between $-\theta_L$ and θ_L and where θ_L is a limiting angle provided by the user(*). The initial value of ϕ is assigned a value randomly selected between 0 and 2π. Following the technique of reference 1, the trajectory intersection with each sampling plane is determined and the coordinates of this intersection are found to either be contained in a cell containing fluid or an empty cell. The scattering at this point is thus related to the cell contents. If the Z coordinate of the particle becomes negative or if the the magnitude of the x or y coordinates

(*) θ_L could of course be as high as $\pi/2$. However, calculations indicated that no significant advantage was gained for $\theta > 30^{\circ}$.

exceeds d, the particle is dropped from further consideration. Once the Z coordinate of the particle exceeds ℓ, the particle is counted. This process is repeated for each trajectory in each configuration in each test.

Program Input

All program input is accomplished using FORTRAN list directed reads. The data may appear anywhere in the field, and when more than one item is specified, the data items are separated by commas. The first items in a data set are the initial gamma ray energy and the cutoff energy in keV. If, as a result of scattering, the gamma ray energy drops below the cutoff value, it is not counted by the detector. The next data items are the dimension parameters of the container interior, ℓ and d, in cm. The next data items are the number of sets, the number of grid lines (5 or 10), and the number of trajectories per distribution. These data are followed by the electron density of the fluid in electron/cm^3, then the distance between sampling planes in cm, and finally the limiting value of θ for particle initialization, in degrees. If additional cases are desired in a single program execution, the entire data set must be repeated for each case. Sample data for a single execution to analyze two cases(*) are illustrated below:

```
662., 200.
30.48, 10.16
10, 5, 400
3.346E+23
1.016
30.

662., 200.
30.48, 10.16
10, 10, 400
3.346E+23
1.016
30.
```

In addition to the data described above, the program also requires a table containing probability versus scattering angle (θ), as described in reference 1. These data should exist as an unformatted file on unit TAPE2.

Program Output

The program generates formatted output on two files. File TAPE6 contains a record of the program input followed by a graphic display of the fluid distribution and the transmission probability for each distribution. An example of this

---

(*) For illustrative purposes, the fluid is assumed to be water. The container dimensions are taken to be 12" x 8" x 8" and the grids considered are (5 x 5) and (10 x 10).

output is contained in Table I, for a 10 x 10 grid. Each distribution from no fluid to 100% fluid is illustrated for each set, with the transmission for each distribution printed below the corresponding display. In the display, each "X" represents a cell containing fluid with embedded blanks representing cells containing air. Following this display for each set, a summary over all sets is output as indicated in Table II. This summary contains the average transmission and its standard deviation for each distribution. File TAPE1 provides a brief summary for each distribution including the number of counts, the number not counted, and the number of particles dropped for x, y, or z coordinates outside the region or for energy below the cutoff value of 200 keV.

In addition to the formatted output, the program generates graphic output presenting the transmission versus fluid content. An example of this output is included as figure 3. The solid line in this figure shows the average transmission over all sets and the dots show the results for each set.

Program Listing

A listing of the program and required subroutines follows. For completeness, the program for generating gamma ray scattering probability vs. θ is also included. Areas of code reflecting limitations of the current program are highlighted by comments.

```
      PROGRAM TDIST2(OUTPUT,INPUT,TAPE6=OUTPUT,TAPE5=INPUT,
     1 TAPE2)
      COMMON PI,RO,ALPHA
      EXTERNAL FUN
      PI = ACOS(-1.)
      RO = 2.818E-13
      ALPHA = 662./511.
      TSCAT = GLEG15(0.,PI,FUN)
      WRITE(6,1)
    1 FORMAT(1H1,10X,5HRZERO,10X,5HALPHA,
     1 2X,27HTOT SCAT CROSS SECTION (PT)///)
      WRITE(6,2) RO,ALPHA,TSCAT
    2 FORMAT(1H0,1PE15.4,1PE22.4,1PE29.4)
      NUM = 180.
      DO 100 I=1,NUM
      THETA = PI*FLOAT(I-1)/180.
      THETAP = THETA + PI/180.
      PSCAT = GLEG15(THETA,THETAP,FUN)
      WRITE(2) THETA,PSCAT
      WRITE(6,3) THETA,PSCAT
    3 FORMAT(2F16.8)
  100 CONTINUE
      CONTINUE
      STOP
      END
```

```
      PROGRAM CRANK(OUTPUT,INPUT,TAPE5=INPUT,TAPE6=OUTPUT,TAPE2,
     1 TAPE1)
      COMMON PI,RO,ALPHA,EZERO,EFINAL,CLEN,CDIAM,DIST,THETAL,XNZW
      DIMENSION TTT(100)
      DIMENSION XX(180),YY(180)
      DIMENSION TRANS(100,10)
      COMMON/RSLT/AVG(11),STD(11)
      DIMENSION IG(100)
    1 FORMAT(1H1,3X,*EZERO,EFINAL,CLEN,CRAD,NSET,NGRD,NGAM,DIST,THETAL*
     1 *,XNZW*//4E16.8//3I10//3E16.8)
    2 FORMAT(1H1,4X,*SUMMARY*///1X,6X,*PERCENT WATER*,3X,*TRANSMISSION*,
     1 6X,*STD. DEV.*//)
    3 FORMAT(5X,3F15.4)
      CALL PSEUDO
      PI = ACOS(-1.)
C
C           GET SCATTERING DISTRIBUTION
C
      DO 10 I=1,180
      READ(2) XX(I),YY(I)
   10 CONTINUE
      SUM = 0.
C
C           GET CUMULATIVE DISTRIBUTION
C
      DO 20 I=1,180
      SUM = SUM + YY(I)
      YY(I) = SUM
      XX(I) = XX(I) + PI/360.
   20 CONTINUE
C
C           NORMALIZE IT
C
      DO 30 I=1,180
      YY(I) = YY(I)/SUM
   30 CONTINUE
      RO = 2.818E-13
C
C           READ INITIAL AND CUTOFF ENERGIES
C
   40 READ(5,*) EZERO,EFINAL
      IF(EOF(5).NE.0) GO TO 999
```

```
C
C          READ LENGTH AND 'RADIUS' OF CONTAINER
C               FOR A RECTANGULAR CONTAINER THE
C               'RADIUS' IS HALF THE WIDTH
C
      READ(5,*) CLEN,CRAD
      CDIAM = 2.*CRAD
C
C          READ THE NUMBER OF SETS, THE NUMBER
C          OF GRIDS, AND THE NUMBER OF GAMMAS
C          PER SET
C
      READ(5,*) NSET,NGRD,NGAM
C
C          READ THE ELECTRON DENSITY OF
C          THE FLUID
C
      READ(5,*) XNZW
C
C          READ THE DISTANCE BETWEEN
C          SAMPLING PLANES
C
      READ(5,*) DIST
C
C          READ THE COLLIMATION ANGLE
C
      READ(5,*) THETAL
      WRITE(6,1) EZERO,EFINAL,CLEN,CRAD,NSET,NGRD,NGAM,DIST,THETAL
     1 ,XNZW
      THETAL = THETAL*PI/180.
      ALPHA = EZERO/511.
      CALL CALPLT(1.5,1.5,-3)
      NGSQ = NGRD*NGRD
      NGRDP = NGRD + 1
      NGRDM = NGRD - 1
C
C          THE 100 LOOP IS OVER SETS
C
      DO 100 K=1,NSET
C
C          ZERO THE GRID
C
```

```
      DO 50 I=1,NGSQ
   50 IG(I) = 0
C
C          THE 80 LOOP IS OVER FLUID CONCENTRATIONS
C
      DO 80 I=1,NGRDM
C
C          RANDOMLY FILL PART OF THE CONTAINER
C
      DO 70 J=1,NGRD
   60 R = RANF(T)
      NR = NGSQ*R + 0.01
      IF(NR.LE.0) NR = 1
      IF(NR.GT.NGSQ) NR = NGSQ
      IF(IG(NR).EQ.1) GO TO 60
      IG(NR) = 1
   70 CONTINUE
C
C          PROCESS THIS CONCENTRATION
C
      CALL CONFIG(IG,I,TTT,TRANS,NSET,NGAM,NGRD,XX,YY)
   80 CONTINUE
      DO 90 I=1,NGSQ
      IG(I) = 1
   90 CONTINUE
C
C          PROCESS THE FILLED CONTAINER
C
      CALL CONFIG(IG,NGRD,TTT,TRANS,NSET,NGAM,NGRD,XX,YY)
  100 CONTINUE
C
C          DISPLAY THE RESULTS
C
      CALL DISP(TTT,TRANS,NSET,NGRD)
C
C          PRINT THE SUMMARY TABLE
C
      WRITE(6,2)
      DO 110 I=1,NGRDP
      PER = (100./FLOAT(NGRD))*FLOAT(I-1)
      WRITE(6,3) PER,AVG(I),STD(I)
  110 CONTINUE
```

```
      GO TO 40
  999 CONTINUE
      CALL CALPLT(0.,0.,999)
      STOP
      END
```

```
      SUBROUTINE CONFIG(IG,IND,TTT,TRANS,NSET,NGAM,NGRD,XX,YY)
C
C          CONFIG IS ENTERED ONCE FOR EACH
C          FLUID CONCENTRATION FOR EACH SET
C
      COMMON PI,RO,ALPHA,EZERO,EFINAL,CLEN,CDIAM,DIST,THETAL,XNZW
      DIMENSION XX(180),YY(180)
      DIMENSION TTT(100)
      DIMENSION NZERO(10),NONE(10),TT(10)
      DIMENSION IG(NGRD,NGRD),KG(100,10)
      DIMENSION LINE(100)
      DIMENSION IZERO(10),TRANS(100,10)
      EXTERNAL FUN
      DATA ISET/0/
      DATA IZERO/10*1H /
    1 FORMAT(1X,11(10A1,1X))
    2 FORMAT(1H0,11(10A1,1X))
    3 FORMAT(1H0,3X,*I,NHIT,NMISS,NNX,NNY,NNZ,NNE*//
     1 7I10//)
    4 FORMAT(1H0,F9.6,2X,10(F9.6,2X))
      CRAD = CDIAM/2.
C
C          STORE CONCENTRATION DATA FOR
C          LATER DISPLAY
C
      NGRDP = NGRD + 1
      NGRDM = NGRD - 1
      NGSQ = NGRD*NGRD
      DO 10 I=1,NGRD
      JND = NGRD*(IND - 1) + I
      DO 10 J=1,NGRD
      KG(JND,J) = IG(I,J)
   10 CONTINUE
C
C          RETURN IF NOT FULL
C
      IF(IND.NE.NGRD) GO TO 180
C
C          LAST CONCENTRATION (FULL),
C          DISPLAY AND PROCESS
C
      ISET = ISET + 1
```

```
      DO 40 J=1,NGRD
      DO 20 KK=1,100
      LINE(KK) = 1H
   20 CONTINUE
      DO 30 KK=1,NGRD
      DO 30 LL=1,NGRD
      JND = NGRD*(KK - 1) + LL
      KND = 10*(KK - 1) + LL
      IF(KG(JND,J).NE.0) LINE(KND) = 1HX
   30 CONTINUE
      IF(J.NE.1) WRITE(6,1) IZERO,LINE
      IF(J.EQ.1) WRITE(6,2) IZERO,LINE
   40 CONTINUE
C
C           TSCAT IS THE TOTAL SCATTERING
C           CROSS SECTION
C
      TSCAT = GLEG15(0.,PI,FUN)
      XNZA = 3.65E+20
      PTW = EXP(-TSCAT*XNZW*DIST)
      PTA = EXP(-TSCAT*XNZA*DIST)
C
C           THE 170 LOOP IS OVER CONCENTRATIONS
C
      DO 170 I=1,NGRDP
      IF(I.NE.NGRDP) TRANS(ISET,I) = 0.
      NHIT = 0
      NMISS = 0
      NSCAT = 0
      NNX = 0
      NNY = 0
      NNZ = 0
      NNE = 0
C
C           THE 150 LOOP IS OVER GAMMAS
C
      DO 150 K=1,NGAM
      THETAS = 0.
      PHIS = 0.
      THETAF = 0.
      PHIF = 0.
      ENER = EZERO
```

```
C
C        XI IS THE INITIAL LOCATION OF
C        THE GAMMA ALONG THE X-AXIS, HERE
C        ASSUMED TO BE UNIFORMLY DISTRIBUTED.
C        THE INITIAL Y-COORDINATE, YI, IS
C        ASSUMED TO BE UNIFORMLY DISTRIBUTED
C        ACROSS THE CONTAINER WALL
C
      XI = CDIAM*RANF(T)
C
C        THETAI IS THE INITIAL THETA-
C        COMPONENT OF THE GAMMA VELOCITY,
C        HERE ASSUMED TO BE UNIFORMLY
C        DISTRIBUTED INSIDE THE
C        COLLIMATION ANGLE. THE INITIAL
C        PHI-COMPONENT IS ASSUMED TO BE
C        UNIFORMLY DISTRIBUTED BETWEEN
C        ZERO AND 2*PI
C
      THETAI = -THETAL/2. + RANF(T)*THETAL
      PHII = 2.*PI*RANF(T)
      THETA = THETAI
      PHI = PHII
      ZI = 0.
      YI = -CRAD + 2.*CRAD*RANF(T)
      XLEN = CDIAM/FLOAT(NGRD)
      ZLEN = CLEN/FLOAT(NGRD)
      XI = XI + DIST*SIN(THETA)*COS(PHI)
      YI = YI + DIST*SIN(THETA)*SIN(PHI)
      ZI = ZI + DIST*COS(THETA)
C
C        IXI AND IZI ARE THE GRID INDICES
C        OF THE CURRENT GAMMA LOCATION
C
      IXI = XI/XLEN + 1.
      IZI = ZI/ZLEN + 1.
C
C        THE FOLLOWING TESTS DETERMINE IF
C        THE GAMMA HAS EXCEEDED THE X OR Y
C        BOUNDARIES OF THE CONTAINER.
C             FOR A CIRCULAR CONTAINER THEY
C             COULD BE REPLACED WITH A TEST
```

```
C                  XI**2 + YI**2 .GT. CRAD**2
C
      IF(IXI.GT.NGRD) GO TO 100
      IF(IXI.LT.1) GO TO 100
      IF(ABS(YI).GT.CRAD) GO TO 110
C
C         STOP IF ENERGY BELOW CUTOFF
C
      IF(ENER.LE.EFINAL) GO TO 120
C
C         STOP IF GAMMA OUTSIDE Z-BOUNDS
C
      IF(IZI.LT.1) GO TO 130
      IF(IZI.GT.NGRD) GO TO 140
   50 CONTINUE
      R = RANF(T)
      IXI = XI/XLEN + 1.
      IZI = ZI/ZLEN + 1.
      PT = PTA
      IF(I.EQ.NGRDP) GO TO 60
      JND = NGRD*(I-1)
C
C         THE EFFECTS OF THE CONTAINER WALLS
C         COULD BE INTRODUCED HERE WITH THE
C         INTRODUCTION OF A 'PT' FOR THE WALL
C         AS A FUNCTION OF XI, YI, AND ZI
C
      IF(KG(JND+IXI,IZI).EQ.1) PT = PTW
   60 CONTINUE
C
C         BRANCH IF NO SCATTER
C
      IF(R.LT.PT) GO TO 90
C
C         THE GAMMA SCATTERED
C
      NSCAT = NSCAT + 1
      R = RANF(T)
      DO 70 KI = 1,179
      IF(YY(KI).LE.R.AND.YY(KI+1).GE.R) GO TO 80
   70 CONTINUE
      KI = 179
```

```
   80 CONTINUE
C
C          DETERMINE SCATTERING ANGLE
C
      THETAS = XX(KI) + (XX(KI+1) - XX(KI))*(R - YY(KI))/
     1 (YY(KI+1) - YY(KI))
      PHIS = 2.*PI*RANF(T)
      THETA = THETA*180./PI
      PHI = PHI*180./PI
      THETAS = THETAS*180./PI
      PHIS = PHIS*180./PI
C
C          GET NEW DIRECTION
C
      CALL POST(THETA,PHI,THETAS,PHIS,THETAF,PHIF)
      THETA = THETAF*PI/180.
      PHI = PHIF*PI/180.
      THETAS = THETAS*PI/180.
C
C          DECREMENT ENERGY
C
      ENER = ENER/(1. + ALPHA*(1. - COS(THETAS)))
   90 CONTINUE
C
C          ADVANCE THE POSITION OF THE GAMMA
C
      XI = XI + DIST*SIN(THETA)*COS(PHI)
      YI = YI + DIST*SIN(THETA)*SIN(PHI)
      ZI = ZI + DIST*COS(THETA)
      IXI = XI/XLEN + 1.
      IZI = ZI/ZLEN + 1.
C
C          REPEAT TESTS TO DROP THE
C          GAMMA FROM FURTHER CONSIDERATION
C
      IF(IXI.GT.NGRD) GO TO 100
      IF(IXI.LT.1) GO TO 100
      IF(ABS(YI).GT.CRAD) GO TO 110
      IF(ENER.LE.EFINAL) GO TO 120
      IF(IZI.LT.1) GO TO 130
      IF(IZI.GT.NGRD) GO TO 140
      GO TO 50
```

```
C
C          A MISS, X OUTSIDE BOUNDS
C
  100 NMISS = NMISS + 1
      NNX = NNX + 1
      GO TO 150
C
C          A MISS, Y OUTSIDE BOUNDS
C
  110 NMISS = NMISS + 1
      NNY = NNY + 1
      GO TO 150
C
C          A MISS, ENERGY BELOW CUTOFF
C
  120 NMISS = NMISS + 1
      NNE = NNE + 1
      GO TO 150
C
C          A MISS, Z OUTSIDE BOUNDS
C
  130 NMISS = NMISS + 1
      NNZ = NNZ + 1
      GO TO 150
  140 CONTINUE
C
C          THE GAMMA REACHED THE FAR CONTAINER
C          WALL, HERE ASSUMED TO BE COUNTED
C
      NHIT = NHIT + 1
  150 CONTINUE
C
C          OUTPUT STATISTICS TO TAPE1
C
      WRITE(1,3) I,NHIT,NMISS,NNX,NNY,NNZ,NNE
      IF(I.EQ.NGRDP) GO TO 160
      TRANS(ISET,I) = FLOAT(NHIT)/FLOAT(NGAM)
      GO TO 170
  160 TTT(ISET) = FLOAT(NHIT)/FLOAT(NGAM)
  170 CONTINUE
      WRITE(6,4) TTT(ISET),(TRANS(ISET,J),J=1,NGRD)
  180 RETURN
      END
```

```
      SUBROUTINE DISP(TTT,TRANS,NSET,NGRD)
C
C            DISP GRAPHICALLY DISPLAYS THE
C            SUMMARY STATISTICS
C
      DIMENSION TTT(100)
      DIMENSION TRANS(100,10)
      DIMENSION X(15),Y(15)
      COMMON/RSLT/AVG(11),STD(11)
      NGRDP = NGRD + 1
      NGRDM = NGRD - 1
      X(1) = 0.
      DO 10 I=1,NGRD
      X(I+1) = X(I) + 100./FLOAT(NGRD)
   10 CONTINUE
      DO 30 I=2,NGRDP
      Y(I) = 0.
      DO 20 J=1,NSET
      Y(I) = Y(I) + TRANS(J,I-1)
   20 CONTINUE
      Y(I) = Y(I)/FLOAT(NSET)
      AVG(I) = Y(I)
   30 CONTINUE
      Y(1) = 0.
      DO 40 J=1,NSET
      Y(1) = Y(1) + TTT(J)
   40 CONTINUE
      Y(1) = Y(1)/FLOAT(NSET)
      AVG(1) = Y(1)
      X(NGRD+2) = 0.
      X(NGRD+3) = 10.
      Y(NGRD+2) = 0.
      Y(NGRD+3) = 0.1
      CALL AXES(0.,0.,0.,10.,X(NGRD+2),X(NGRD+3),1.,0.,
     1 7H% WATER,0.2,-7)
      CALL AXES(0.,0.,90.,10.,Y(NGRD+2),Y(NGRD+3),1.,0.,
     1 12HTRANSMISSION,0.2,12)
      CALL LINPLT(X,Y,NGRDP,1,0,0,0,0)
      DO 50 J=1,NGRDM
      XP = 100*J/(10*NGRD)
      DO 50 I=1,NSET
      YP = TRANS(I,J)/0.1
```

```
      CALL PNTPLT(XP,YP,21,1)
   50 CONTINUE
      DO 80 I=2,NGRDP
      STD(I) = 0.
      DO 60 J=1,NSET
      STD(I) = STD(I) + (AVG(I) - TRANS(J,I-1))**2
   60 CONTINUE
      STD(I) = STD(I)/FLOAT(NSET)
      STD(I) = SQRT(STD(I))
      STD(1) = 0.
      DO 70 J=1,NSET
      STD(1) = STD(1) + (AVG(1) - TTT(J))**2
   70 CONTINUE
      STD(1) = STD(1)/FLOAT(NSET)
      STD(1) = SQRT(STD(1))
   80 CONTINUE
      RETURN
      END
```

```
SUBROUTINE POST(T1,P1,T2,P2,T3,P3)
DIMENSION A(4,4)
CALL UROTOC(1,T2,A)
CALL UAPPLY(0.,0.,1.,A,X,Y,Z)
CALL UROTOC(3,P2,A)
CALL UAPPLY(X,Y,Z,A,X,Y,Z)
CALL UROTOC(1,T1,A)
CALL UAPPLY(X,Y,Z,A,X,Y,Z)
CALL UROTOC(3,P1,A)
CALL UAPPLY(X,Y,Z,A,X,Y,Z)
P3=ATAN(X/Y)*180./3.14159265
IF (X.LT.0..AND.Y.LT.0.) P3=360.-P3
IF (X.LT.0..AND.Y.GE.0.) P3=180.-P3
IF (X.GE.0..AND.Y.GE.0.) P3=90.+P3
IF (X.GE.0..AND.Y.LT.0.) P3=-P3
R=SQRT(X*X+Y*Y)
IF (Y.LT.0.) R=-R
T3=ATAN(R/Z)*180./3.14159265
IF (R.LT.0..AND.Z.LT.0.) T3=180.-T3
IF (R.LT.0..AND.Z.GT.0.) T3=-T3
IF (R.GE.0..AND.Z.GE.0.) T3=T3
IF (R.GE.0..AND.Z.LT.0.) T3=180.+T3
RETURN
END
```

```
      SUBROUTINE UAPPLY (X,Y,Z, A, U,V,W)
C
C POST-MULTIPLIES A POINT VECTOR (X Y Z) BY 4*4
C TRANSFORMATION MATRIX A TO GIVE (U V W).
C
      REAL A(4,4)
C H IS THE HOMOGENEOUS CO-ORDINATE, WHICH IS USED
C TO EFFECT VARIOUS PROJECTIONS.  IN THIS ROUTINE,
C MATRIX A IS ASSUMED TO PRODUCE ONLY AFFINE
C TRANSFORMATIONS, WHICH MEANS THE LAST COLUMN OF
C MATRIX A IS (0 0 0 1), SO H IS UNITY.  BUT IT,S
C HERE IN CASE ONE WISHES TO CHANGE IT...
C (UNCOMMENT NEXT LINE AND DIVIDE X, Y, AND Z BY H.)
C     H = X*A(1,4) + Y*A(2,4) + Z*A(3,4) + A(4,4)
      U = (X*A(1,1) + Y*A(2,1) + Z*A(3,1) + A(4,1))
      V = (X*A(1,2) + Y*A(2,2) + Z*A(3,2) + A(4,2))
      W = (X*A(1,3) + Y*A(2,3) + Z*A(3,3) + A(4,3))
      RETURN
      END
```

```
      SUBROUTINE UROTOC (IAXIS, ANGLE, A)
C
C CONSTRUCTS A 4*4 MATRIX THAT PERFORMS A ROTATION
C ABOUT X-, Y-, OR Z-AXIS OF ANY ANGLE.
C
C INPUTS --
C   IAXIS     1=X  2=Y  3=Z
C   ANGLE     AMOUNT OF ROTATION, IN CURRENT UNITS.  MEASURED
C             POSITIVE BY RIGHT-HAND RULE.
C
      REAL A(4,4)
      COMMON /TORADS/ ANGFAC
      DATA ANGFAC /.01745329252/
C
      IF (IAXIS .LE. 0 .OR. IAXIS .GE. 4)  GO TO 40
      CALL UCLR (A)
      A(4,4) = 1.
      RAD = ANGFAC * ANGLE
      COSANG = COS (RAD)
      SINANG = SIN (RAD)
      GO TO (10, 20, 30), IAXIS
   10 A(1,1) = 1.
      A(2,2) = COSANG
      A(3,2) = - SINANG
      A(2,3) = SINANG
      A(3,3) = COSANG
      RETURN
   20 A(2,2) = 1.
      A(1,1) = COSANG
      A(3,1) = SINANG
      A(1,3) = - SINANG
      A(3,3) = COSANG
      RETURN
   30 A(3,3) = 1.
      A(1,1) = COSANG
      A(2,1) = - SINANG
      A(1,2) = SINANG
      A(2,2) = COSANG
      RETURN
   40 CALL UIDENT (A)
      RETURN
      END
```

```
      SUBROUTINE UCLR (A)
C
C CLEARS 4*4 MATRIX A TO ZEROS.
C
      REAL A(4,4)
      DO 10 I=1,4
        DO 10 J=1,4
   10     A(I,J) = 0.
      RETURN
      END
```

```
      SUBROUTINE UIDENT (A)
C
C MAKES 4*4 IDENTITY MATRIX A.
C
      REAL A(4,4)
      CALL UCLR (A)
      DO 10 I=1,4
   10   A(I,I) = 1.
      RETURN
      END
```

```
      SUBROUTINE FUN(F,N)
      DIMENSION F(1)
      COMMON PI,RO,ALPHA
      DO 10 I=1,N
      THETA = F(I)
      ANS = 2.*PI*RO**2*SIN(THETA)
      TMP1 = 1. + ALPHA*(1. - COS(THETA))
      TMP2 = 1. + COS(THETA)**2
      TMP3 = (1. - COS(THETA))**2
      ANS = ANS*(1./TMP1)**2*(TMP2/2.)
     1 *(1. + ALPHA**2*TMP3/(TMP2*TMP1))
      F(I) = ANS
   10 CONTINUE
      RETURN
      END
```

```
      FUNCTION GLEG15(A,B,FUNCT)
C
C FUNCTION GLEG15 EVALUATES INTEGRALS OF THE FORM
C
C          B
C     INTEGRAL [G(X)] DX
C          A
C
C WHERE A AND B ARE FINITE USING 15 POINT GUASS-LEGENDRE QUADRATURE.
C
C USE
C
C     VALUE=GLEG15(A,B,FUNCT)
C
C            A......THE LOWER LIMIT OF INTEGRATION. MUST BE FINITE.
C
C            B......THE UPPER LIMIT OF INTEGRATION. MUST BE FINITE.
C
C            FUNCT..THE NAME OF AN EXTERNAL SUBROUTINE TO CALCULATE THE
C                   VALUE OF G(X) AT SPECIFIED POINTS.
C                   SUBROUTINE FUNCT(F,N)
C                              F..ON INPUT THE N VALUES OF X
C                                 ON OUTPUT THE N VALUES G(X)
C                                 THAT IS, F(I)=G[F(I)] I=1,...,N
C                              N..THE NUMBER OF FUNCTION EVALUATIONS.
C                   THE NAME GIVEN FUNCT MUST APPEAR IN AN EXTERNAL
C                   STATEMENT IN THE CALLING PROGRAM
      DIMENSION W(8),R(8),F(15)
      DATA R/0.             ,.201194093997434,.394151347077563,
     *       .570972172608538,.724417731360170,.848206583410427,
     *       .937273392400705,.987992518020485/
      DATA W/.202578241925561,.198431485327111,.186161000015562,
     *       .166269205816993,.139570677926154,.107159220467171,
     *       .703660474881081E-01,.307532419961172E-01/
      C=(B-A)*.5
      D=(B+A)*.5
      F(1)=D
      DO 10 I=2,8
      S=R(I)*C
      F(I)=D+S
   10 F(I+7)=D-S
      CALL FUNCT(F,15)
```

```
      D=W(1)*F(1)
      DO 20 I=2,8
   20 D=D+W(I)*(F(I)+F(I+7))
      GLEG15=C*D
      RETURN
      END
```

DISCUSSION

As seen from Table III, the gamma ray transmissions for uniform planar source and the uniform linear source for various levels of fluid content in the container are essentially the same if the fluid distribution is assumed to be random. It might thus appear that a uniform linear source passing vertically through the center of the one side of the container would suffice for fluid content measurement. However, if the fluid is concentrated against one of the container walls or is confined to corner regions due to centrifugal forces, the linear source will not give a correct indication about the fluid content of the container. It is therefore concluded that a uniform planar Cs^{137} source, coupled with an appropriate planar detector on the opposite face, would provide a more appropriate--though computationally less efficient--basis for developing a universal nonintrusive monitor for liquid contents in sealed vessels.

For the source/detector geometry illustrated in figure 1 and for a measurement accuracy of $\lessdot$ 10 percent, a planar Cs^{137} gamma ray source of strength 10 μc will be adequate. Such an arrangement is expected to be particularly sensitive when the fluid content falls below 50 percent level.

CONCLUDING REMARKS

A nonintrusive nuclear monitor for liquid contents in sealed vessels has been described. Illustrative test studies indicate that the technique is capable of providing accurate (< $\pm$10%) information about the residual fluid content in sealed vessels. The monitor would be useable in all environments--ranging from 0-g of space to high-g of fast maneuvering aircraft.

REFERENCES

1. Singh, J. J., Davis, W. T., and Mall, G. H.: Development of a Nuclear Technique for Monitoring Water Levels in Pressurized Vessels, NASA TM-85651, 1983.

2. Evans, Robley D.: The Atomic Nucleus. McGraw-Hill Company, Inc., c. 1955.

3. Goldstein, Herbert: Fundamental Aspects of Reactor Shielding. Addison-Wesley Publishing Company, Inc., c. 1959.

4. Alder, B.; Fernbach, S.; and Rotenberg, M., eds.: Methods in Computational Physics, Vol. 1, Academic Press, Inc., 1963.

Table I. Sample Formatted Display Output for Planar Source/Planar Detector Configuration

EZERO	0.6620E+03	CRAD	0.1016E+02	NGAM	400
EFINAL	0.2000E+03	NSET	10	DIST	0.1016E+01
CLEN	0.3048E+02	NGRAD	10	THETAL	0.3000E+02
				XNZW	0.3346E+24



Fluid content (%) →	0	10	20	30	40	50	60	70	80	90	100
Case 1 — Computed transmission →	.747500	.602500	.540000	.480000	.407500	.350000	.305000	.225000	.215000	.170000	.097500
Case 2 — Computed transmission →	.762500	.652500	.517500	.417500	.370000	.295000	.260000	.187500	.190000	.152500	.110000
Case 3 — Computed transmission →	.752500	.662500	.560000	.475000	.340000	.317500	.305000	.237500	.185000	.157500	.130000

Table II. Sample Formatted Summary Output for Planar Source/Detector Configuration.

EZERO	0.6620E+03	NGRAD	10
EFINAL	0.2000E+03	NGAM	400
CLEN	0.3048E+02	DIST	0.1016E+01
CRAD	0.1016E+02	THETAL	0.3000E+02
NSET	10	XNZW	0.3346E+24

Percent Water	Gamma Ray Transmission	Std. Dev.
0.0000	0.7635	0.0221
10.0000	0.6420	0.0324
20.0000	0.5350	0.0317
30.0000	0.4538	0.0264
40.0000	0.3795	0.0220
50.0000	0.3090	0.0239
60.0000	0.2705	0.0302
70.0000	0.2263	0.0228
80.0000	0.1908	0.0177
90.0000	0.1583	0.0107
100.0000	0.1225	0.0209

Table III. Summary of Transmitted Gamma Ray Intensities for Various Fluid Contents.

(a) Uniform Planar Source

Percent Water	Transmission		$R^{(*)}$	
(%)	Random Fluid Distribution	Non-Random Fluid Distribution	Random Fluid Distribution	Non-Random Fluid Distribution
0.0000	0.7635±0.0221	0.7560±0.0198	6.233±0.200	5.458±0.133
10.0000	0.6420±0.0324	0.6002±0.0175	5.241±0.221	4.334±0.136
20.0000	0.5350±0.0317	0.4777±0.0193	4.367±0.230	3.449±0.147
30.0000	0.4538±0.0264	0.3890±0.0271	3.704±0.229	2.809±0.177
40.0000	0.3795±0.0220	0.3103±0.0140	3.098±0.229	2.240±0.152
50.0000	0.3090±0.0239	0.2670±0.0131	2.522±0.248	1.928±0.156
60.0000	0.2705±0.0302	0.2343±0.0205	2.208±0.282	1.692±0.194
70.0000	0.2263±0.0228	0.2025±0.0165	1.347±0.271	1.462±0.188
80.0000	0.1908±0.0177	0.1755±0.0149	1.558±0.263	1.267±0.192
90.0000	0.1583±0.0107	0.1465±0.0159	1.292±0.238	1.058±0.215
100.0000	0.1228±0.0209	0.1385±0.0148	1.000±0.341	1.000±0.214

$^{(*)}R = \dfrac{\text{Transmission for x \% Water}}{\text{Transmission for 100\% Water}}$

Table III. Summary of Transmitted Gamma Ray Intensities for Various Fluid Contents.

(b) Uniform Linear Source

Percent Water	Transmission		R(*)	
(%)	Random Fluid Distribution	Non-Random Fluid Distribution	Random Fluid Distribution	Non-Random Fluid Distribution
0.0000	0.8732±0.0201	0.8682±0.0134	5.940±0.158	5.788±0.115
10.0000	0.7407±0.0289	0.7080±0.0184	5.039±0.174	4.720±0.126
20.0000	0.6327±0.0281	0.5727±0.0153	4.304±0.179	3.818±0.127
30.0000	0.5277±0.0303	0.4560±0.0243	3.590±0.192	3.040±0.153
40.0000	0.4455±0.0246	0.3653±0.0203	3.031±0.190	2.435±0.156
50.0000	0.3770±0.0320	0.3070±0.0177	2.565±0.220	2.047±0.158
60.0000	0.3295±0.0182	0.2660±0.0174	2.241±0.190	1.773±0.165
70.0000	0.2613±0.0196	0.2293±0.0227	1.778±0.210	1.529±0.159
80.0000	0.2088±0.0280	0.1955±0.0152	1.420±0.269	1.303±0.178
90.0000	0.1785±0.0177	0.1640±0.0117	1.214±0.234	1.093±0.171
100.0000	0.1470±0.0198	0.1500±0.0150	1.000±0.269	1.000±0.200

(*) $R = \dfrac{\text{Transmission for x \% Water}}{\text{Transmission for 100\% Water}}$



FIGURE 1. - SCHEMATIC DIAGRAM FOR THE UNIFORM PLANAR SOURCE/PLANAR DETECTOR CONFIGURATION. THE CONTAINER IS ASSUMED TO BE A $\ell \times 2d \times 2d$ RECTANGULAR BOX FOR COMPUTATIONAL PURPOSES.

Page missing in original document.



FIGURE 3. - GAMMA RAY TRANSMISSION AS A FUNCTION OF THE FLUID CONTENT OF THE VESSEL FOR PLANAR SOURCE/DETECTOR CONFIGURATION.

1. Report No. NASA TM-85754	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle A Nonintrusive Nuclear Monitor for Measuring Liquid Contents in Sealed Vessels		5. Report Date February 1984
		6. Performing Organization Code 505-33-53-14
7. Author(s) Jag J. Singh and Gerald H. Mall		8. Performing Organization Report No.
9. Performing Organization Name and Address NASA Langley Research Center Hampton, VA 23665		10. Work Unit No.
		11. Contract or Grant No.
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546		13. Type of Report and Period Covered Technical Memorandum
		14. Sponsoring Agency Code

15. Supplementary Notes

Jag J. Singh: Langley Research Center, Hampton, Virginia
Gerald H. Mall: Computer Sciences Corporation, Hampton, Virginia

16. Abstract

This report describes a nonintrusive nuclear technique for monitoring fluid contents in sealed vessels, regardless of the fluid distribution inside the vessels. The technique is applicable to all-g environments. It is based on the differences in Cesium-137 gamma ray attenuation coefficients in air and the test liquids.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Sealed Fluid Container 0-g Environment Nuclear Monitor Compton Scattering Cross Section Multiple Scattering	Unclassified - Unlimited Subject Category 35

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price
Unclassified	Unclassified	32	A03